
This session is devoted to symbolic computation of the first Lyapunov coefficient

l1 for Andronov-Hopf bifurcation in planar systems

ẋ = F (x, α), x ∈ R
2, α ∈ R. (0.1)

The algorithm for computing l1 can be summarized as follows.

Step 0: Find a critical parameter value α0 for which the system has an equilibrium
x0 with eigenvalues λ1,2 = ±iω0, ω0 > 0. This can be achievd, for example, by solving
the nonlinear system

{

F (x, α) = 0,
Tr Fx(x, α) = 0

for (x0, α0) and verifying that det Fx(x0, α0) = ω2
0 > 0.

Step 1: Compute complex vectors q, p ∈ C
2 such that

Aq = iω0q, AT p = −iω0p, 〈p, q〉 = 1,

where A = Fx(x0, α0) and 〈p, q〉 = p̄1q1 + p̄2q2.

Step 2: Compose the expression

H(z, z̄) = 〈p, F (x0 + zq + z̄q̄, α0)〉

and find the coefficients g20, g11, and g21 in the formal Taylor series

H(z, z̄) = iω0z +
∑

j+k≥2

1

j!k!
gjkz

j z̄k.

Step 3: Compute the first Lyapunov coefficient:

l1 =
1

2ω2
0

Re (ig20g11 + ω0g21).

If l1 < 0 (l1 > 0), then the equilibrium x0 is stable (unstable) at α0 and a unique
stable (unstable) limit cycle bifurcates from x0 in (0.1) as α passes α0, provided that

d

dα
Tr Fx(x0(α), α)

∣

∣

∣

∣

∣

α=0

6= 0.

In this session we use the symbolic manipulation software maple that can be
started, for example, by the menu command

Start->Standard Applications->Mathematics->Maple 15->

Classic Worksheet Maple 15

1 Andronov-Hopf bifurcation in the Brusselator

Consider the following chemical model (Brusselator):

{

ẋ1 = A − (B + 1)x1 + x2
1x2 ≡ F1(x1, x2, A, B),

ẋ2 = Bx1 − x2
1x2 ≡ F2(x1, x2, A, B).

(1.1)

We fix A > 0 and take B as a bifurcation parameter. The following sequence of maple

commands proves that at B = 1 + A2 the system exhibits a supercritical Andronov-
Hopf bifurcation. Note that output details can depend on the maple version and
computer used.

> with(linalg):

This command allows us to use the maple linear algebra package.

> F[1]:=A-(B+1)*X[1]+X[1]^2*X[2];

F1 := A − ( B + 1 ) X1 + X1
2 X2

> F[2]:=B*X[1]-X[1]^2*X[2];

F2 := B X1 − X1
2 X2

> J:=jacobian([F[1],F[2]],[X[1],X[2]]);

J :=







−B − 1 + 2 X1 X2 X1
2

B − 2 X1 X2 −X1
2







> K:=transpose(J);

K :=







−B − 1 + 2 X1 X2 B − 2 X1 X2

X1
2 −X1

2







By these commands we input the right-hand sides of (1.1) into maple and com-
pute its Jacobian matrix J and its transpose K.

> sol:=solve({F[1]=0,F[2]=0},{X[1],X[2]});

sol :=
{

X1 = A, X2 =
B

A

}

> assign(sol);

3



By this we have found the unique equilibrium of (1.1):

X =
(

A,
B

A

)T

as function of A and B.

> T:=eval(trace(J));

T := B − 1 − A2

> diff(T,B);
1

> sol:=solve({T=0},{B});

sol := {B = 1 + A2 }

> assign(sol);

These commands show that at B = 1+A2 the trace of J evaluated at the equilib-
rium X vanishes: tr J = 0. This means that J has two eigenvalues with λ1 + λ2 = 0.
Moreover, the derivative of tr J with respect to the parameter B is always nonzero
(equals to 1).

> assume(A>0);

> omega:=sqrt(det(J));

ω := A˜

Thus, at B = 1 + A2 the Brusselator has the equilibrium

X =

(

A,
1 + A2

A

)T

with purely imaginary eigenvalues λ1,2 = ±iω, ω =
√

det(J) = A > 0. Therefore,
the Andronov-Hopf bifurcation condition is satisfied. To apply the theoretical results
on the bifurcation, we have to verify that

(i) the eigenvalues cross the imaginary axes with a nonzero velocity with respect
to parameter B;

(ii) the first Lyapunov coefficient l1 does not vanish.

The first genericity condition has been already verified, since Tr J = 2Re λ1,2 and
the derivative of the trace with respect to B is equal to 1. Moreover, the equilibrium

is stable for B < 1 + A2 and unstable for B > 1 + A2. The rest of the section is
devoted to the computation of l1.

> ev:=eigenvects(map(eval,J));

ev :=

[

A˜ I, 1,

{[

1 , −
−A˜I + A˜2

A˜2

]}]

,

[

−I A˜, 1,

{[

1, −
A˜I + A˜2

A˜2

]}]

> q:=ev[1][3][1];

q :=

[

1, −
−A˜I + A˜2

1 + A˜2

]

> q:=evalm(1/q[2]*q): q:=map(simplify,map(evalc,q));

q :=

[

−
A˜(A˜ + I)

1 + A˜2
, 1

]

> et:=eigenvects(map(eval,K));

et :=

[

A˜ I , 1,

{[

A˜ I + A˜2

A˜2
, 1

]}]

,

[

−I A˜, 1,

{[

−A˜I + A˜2

A˜2
1

]}]

> p:=et[2][3][1];

p :=

[

−A˜I + A˜2

A˜2
, 1

]

These commands compute

q =

(

−
iA + A2

1 + A2
, 1

)T

, p =

(

−iA + A2

A2
, 1

)T

,

the critical eigenvectors of the Jacobian matrix J and its transpose, so that

Jq = iωq, JT p = −iωp.

> s1:=simplify(evalc(conjugate(p[1])*q[1]+conjugate(p[2])*q[2]));

s1 := −
2(−1 + A˜ I)

1 + A˜2

> p:=evalm(1/conjugate(s1)*p): p:=map(simplify,map(evalc,p));

p :=









−1

2
I ( 1 + A˜2 )

A˜
,
1

2
−

1

2
I A˜









> simplify(evalc(conjugate(p[1])*q[1]+conjugate(p[2])*q[2]));

1

By the commands above, we achieve the normalization 〈p, q〉 = 1, finally taking

p =

(

−
i(1 + A2)

2A
,

1 − iA

2

)T

.

> F[1]:=A-(B+1)*x[1]+x[1]^2*x[2];

F1 := A˜ − ( 2 + A˜2 ) x1 + x1
2 x2

> F[2]:=B*x[1]-x[1]^2*x[2];

F2 := ( 1 + A˜2 ) x1 − x1
2 x2

> x[1]:=evalc(X[1]+z*q[1]+z1*conjugate(q[1]));

x1 := A˜ −
z A˜2

1 + A˜2
−

z1 A˜2

1 + A˜2
+
(

−
z A˜

1 + A˜2
+

z1 A˜

1 + A˜2

)

I

> x[2]:=evalc(X[2]+z*q[2]+z1*conjugate(q[2]));

x2 :=
1

A˜
+ A˜ + z + z1

> H:=simplify(evalc(conjugate(p[1])*F[1]+conjugate(p[2])*F[2]));

H := −
1

2
(−z1 2 A˜2 z + 2 z z1 A˜ + 3 z2 A˜2 z1 + I z1 2 − 2 z1 2 A˜3

− 2 z2 A˜3 + I z2 − z2 A˜ − z1 2 A˜ + z3 A˜2 − 3 z1 3 A˜2 + 3 A˜4 z2 z1

+ 3 A˜4 z1 2 z − 2 z A˜5 z1 + 2 I z2 A˜2 − 2 I z z1 − 2 I z A˜

+ I z1 2 A˜4 + I z3 A˜ + I z1 3 A˜ − 2 I z A˜5 − 3 I z1 3 A˜3 + I z3 A˜3

− 4 I z A˜3 + I z2 A˜4 + 2 I z1 2 A˜2 − I z2 A˜ z1 − 5 I z1 2 A˜3 z

− I z1 2 A˜ z + 2 I z A˜4 z1 − I z2 A˜3 z1 + A˜4 z3 + A˜4 z1 3

− z1 2 A˜5 − z2 A˜5)
/

( 1 + A˜2 )2

By means of these commands, we set x = X + zq + z̄q̄ and compose the function

H(z, z̄) = 〈p, F (X + zq + z̄q̄, A, 1 + A2)〉.

(In our maple input, z1 stands for z̄.)

> g[2,0]:=simplify(2*evalc(coeftayl(H,[z,z1]=[0,0],[2,0])));

g2,0 := A˜ − I

> g[1,1]:=simplify(evalc(coeftayl(H,[z,z1]=[0,0],[1,1])));

g1,1 :=
( A˜2 − 1 ) ( A˜− I)

1 + A˜2

> g[2,1]:=simplify(2*evalc(coeftayl(H,[z,z1]=[0,0],[2,1])));

g2,1 := −
A˜ ( 3 A˜ − I )

1 + A˜2

The above commands compute the coefficients gjk of the Taylor expansion of
H(z, z̄) at (z, z̄) = (0, 0),

H(z, z̄) = iωz +
∑

2≤j+k≤3

1

j!k!
gjkz

j z̄k + O(|z|4),

giving

g20 = A − i, g11 =
(A − i)(A2 − 1)

1 + A2
, g21 =

A(i − 3A)

1 + A2
.

> l[1]:=simplify(1/(2*omega^2)*Re(I*g[2,0]*g[1,1]+omega*g[2,1]));

l1 := −
2 + A˜2

2A˜ ( 1 + A˜2 )

This final command computes the first Lyapunov coefficient

l1 =
1

2ω2
Re(ig20g11 + ωg2,1) = −

2 + A2

2A(1 + A2)
< 0,

and allows us to verify that it is negative for all A > 0.

2 Andronov-Hopf bifurcation in a predator-prey

model

Compute the first Lyapunov coefficient for the Andronov-Hopf bifurcation in the
system:

ẋ1 = rx1(1 − x1) −
cx1x2

α + x1

,

ẋ2 = −dx2 +
cx1x2

α + x1

.

The system describes dynamics of a simple predator-prey ecosystem. Here x1 and x2

are (scaled) population numbers, and r, c, d, and α are positive parameters charac-
terizing the behavior of isolated populations and their interaction. Consider α as a
control parameter and assume c > d.

2.1 Steps of the analysis

(i) Consider an orbitally equivalent polynomial system
{

ẋ1 = rx1(α + x1)(1 − x1) − cx1x2 ≡ F1(x1, x2, α)
ẋ2 = −αdx2 + (c − d)x1x2 ≡ F2(x1, x2, α).

(2.1)

(ii) Show that at the parameter value

α0 =
c − d

c + d

the system (2.1) has an equilibrium x0 with eigenvalues λ1,2 = ±iω, ω > 0, where

ω2 =
rc2d(c − d)

(c + d)3
.

(iii) Write the Jacobian matrix of (2.1) at x0 for α0 in the form 1

A =







0 − cd
c + d

ω2(c + d)
cd

0





 ,

and find complex eigenvectors q, p of A and AT corresponding to iω and −iω, respec-
tively, and satisfying 〈p, q〉 = 1.

(iv) Compute the coefficients g20, g11, and g21 of the Taylor expansion of the fun
ction g = 〈p, F (x0 + zq + z̄q̄, α0)〉 with respect to z, z̄ at z = z̄ = 0, and check that

l1 =
1

2ω2
Re(ig20g11 + ωg21) < 0.

Conclude that a unique and stable limit cycle bifurcates from the equilibrium via the
Andronov-Hopf bifurcation for α < α0

2.2 maple session

> with(linalg):

> F[1]:=r*X[1]*(alpha+X[1])*(1-X[1])-c*X[1]*X[2];

F1 := r X1 (α + X1) (1 − X1) − c X1 X2

> F[2]:=-alpha*d*X[2]+(c-d)*X[1]*X[2];

F2 := −α d X2 + ( c − d ) X1 X2

1It is always useful to express the Jacobian matrix using ω, since this simplifies epressions for the

eigenvectors.

> J:=jacobian([F[1],F[2]],[X[1],X[2]]);

J :=
[r (α + X1) (1 − X1) + r X1 (1 − X1) − r X1 (α + X1) − c X2 ,−c X1]

[( c − d ) X2 ,−α d + ( c − d ) X1]

> K:=transpose(J);

K :=
[

r (α + X1) (1 − X1) + r X1 (1 − X1) − r X1 (α + X1) − c X2 ,

( c − d ) X2

]

[−c X1 ,−α d + ( c − d ) X1]

> sol:=solve({F[1]=0,F[2]=0,trace(J)=0},{X[1],X[2],alpha});

sol := {X2 = 0, X1 = 0, α = 0} ,

{

X1 =
d

c + d
, α =

c − d

c + d
, X2 =

r c

( c + d )2

}

,

{

X2 = 0, X1 = 1, α =
−r + c − d

r + d

}

,

{

X2 = 0, X1 =
r + c

r
, α = −

r + c

r

}

> sol:=sol[2];

sol :=

{

X1 =
d

c + d
, α =

c − d

c + d
, X2 =

r c

( c + d )2

}

> assign(sol);

> assume(r>0,c>0,d>0,c>d);

> J:=map(simplify,map(eval,J));

J :=













0 −
c˜ d˜

c˜ + d˜

( c˜ − d˜ ) r˜ c˜

( c˜ + d˜ )2
0













> DD:=simplify(det(J));

DD :=
c˜2 d˜ ( c˜ − d˜ ) r˜

( c˜ + d˜ )3

> A:=matrix(2,2,[0,J[1,2],omega^2*J[2,1]/DD,0]);

A :=













0 −
c˜ d˜

c˜ + d˜

ω2 ( c˜ + d˜ )

c˜ d˜
0













> det(A);

ω2

> assume(omega>0);

> ev:=eigenvects(map(eval,A));

ev :=

[

I ω˜, 1,

{[

I c˜ d˜

ω˜ ( c˜ + d˜ )
, 1

]}]

,

[

−I ω˜, 1,

{[

−
I c˜ d˜

ω˜ ( c˜ + d˜ )
, 1

]}]

> q:=ev[1][3][1];

q :=

[

I c˜ d˜

ω˜ ( c˜ + d˜ )
, 1

]

> et:=eigenvects(map(eval,transpose(A)));

et :=

[

I ω˜, 1,

{[

−
I ( c˜ + d˜ ) ω˜

c˜ d˜
, 1

]}]

,

[

−I ω˜, 1,

{[

I ( c˜ + d˜ ) ω˜

c˜ d˜
, 1

]}]

> P:=et[2][3][1];

P :=

[

I ( c˜ + d˜ ) ω˜

c˜ d˜
, 1

]

> s:=simplify(evalc(conjugate(P[1])*q[1]+conjugate(P[2])*q[2]));

s := 2

> p:=evalm(1/conjugate(s)*P);

p :=









1

2
I ω˜ ( c˜ + d˜ )

c˜ d˜
,

1

2









> simplify(evalc(conjugate(p[1])*q[1]+conjugate(p[2])*q[2]));

1

> F[1]:=r*x[1]*(alpha+x[1])*(1-x[1])-c*x[1]*x[2];

F1 := r˜ x1

(

c˜ − d˜

c˜ + d˜
+ x1

)

(1 − x1) − c˜x1 x2

> F[2]:=-alpha*d*x[2]+(c-d)*x[1]*x[2];

F2 := −
( c˜ − d˜ ) d˜x2

c˜ + d˜
+ ( c˜ − d˜ ) x1 x2

> x[1]:=evalc(X[1]+z*q[1]+z1*conjugate(q[1]));

x1 :=
d˜

c˜ + d˜
+ I

(

z c˜ d˜

ω˜ ( c˜ + d˜ )
−

z1 c˜ d˜

ω˜ ( c˜ + d˜ )

)

> x[2]:=evalc(X[2]+z*q[2]+z1*conjugate(q[2]));

x2 :=
r˜ c˜

( c˜ + d˜ )2
+ z + z1

> H:=simplify(evalc(conjugate(p[1])*F[1]+conjugate(p[2])*F[2])):

> g[2,0]:=simplify(2*evalc(coeftayl(H,[z,z1]=[0,0],[2,0])));

g2,0 := −( c˜2 ω˜ − I c˜2 d˜ + 2 ω˜ d˜ c˜ + I d˜2 r˜ + I d˜3 + ω˜ d˜2 ) c˜
/(

ω˜ ( c˜ + d˜ )2
)

> g[1,1]:=simplify(evalc(coeftayl(H,[z,z1]=[0,0],[1,1])));

g1,1 :=
I r˜ d˜2 c˜

ω˜ ( c˜ + d˜ )2

> g[2,1]:=simplify(2*evalc(coeftayl(H,[z,z1]=[0,0],[2,1])));

g2,1 := −3
r˜ d˜2 c˜2

( c˜ + d˜ )2 ω˜2

> l[1]:=factor(1/(2*omega^2)*Re(I*g[2,0]*g[1,1]+omega*g[2,1]));

l1 := −
c˜2 d˜2 r˜

ω˜3 ( c˜ + d˜ )2

3 Additional problems

• Verify that the complex pair of eigenvalues of the positive equilibrium in (2.1)
crosses the imaginary axes with nonzero velocity at the Hopf bifurcation.

• Check that each of the following systems has an equilibrium that exhibits a
Hopf bifurcation at some value of parameter α, and compute the first Lyapunov
coefficient. Verify also that the eigenvalues cross the imaginary axis at nonzero
velocity.

(a) Rayleigh’s equation:
ẍ + ẋ3 − 2αẋ + x = 0

(Hint: Introduce y = ẋ and rewrite the equation as a system of two differential
equations.)

(b) Van der Pol’s oscillator:

ÿ − (α − y2)ẏ + y = 0;

(c) Bautin’s example:

{

ẋ = y

ẏ = −x + αy + x2 + xy + y2;

(d) Advertising diffusion model by G. Feichtinger :

{

ẋ1 = α[1 − x1x
2
2 + A(x2 − 1)]

ẋ2 = x1x
2
2 − x2.

• Analyze Hopf bifurcation in the predator-prey model by Bazykin&Khibnik
(1981):

ẋ =
x2(1 − x)

n + x
− xy,

ẏ = −γy(m − x).

where γ, n, and m are positive parameters.

(1) Find a realationship between parameters m and n corresponding to a Hopf
bifurcation in the system.

(2) Compute the first Lyapunov coefficient l1 along the curve in the (m, n)-plane
found at step (1) and show that it vanishes at

(m, n) =
(

1

4
,
1

8

)

.

